```
<<
 "C:\\Users\\Jamal\\Desktop\\main\\Research\\Codes\\IDOs\\ArtesteinsReduction\\
   TenReS.m"
```

```
Package TenReS version 0.2.4
written by Clemens G. Raab
Copyright 2016, RICAM, Austrian Academy of Sciences
```

A few definitions need to be made by the user. Type ?CoeffQ, ?Specialization, and ?CyclicModules for more information.

Basic definitions

Reduction system for IDOLS

Auxiliary functions

Examples of subsections 3.3 and 3.4 in TDS paper

First we need to introduce to the package the coefficients A_0, A_1, H_1, H_2, Φ, Θ, and $\tilde{\Theta}$ which we use in our examples as elements of the coefficient ring R, the parameter h and the notation $\delta = \sigma_{1,h}$.

```
MemberQR[A0 | A1 | H1 | H2 | Φ | Θ | ΘΘ] := True
MemberQZ[h] := True
δ := S[1, h]
```

Ex.8) Variation of constants with square coefficients

Consider the inhomogeneous differential system

$$\dot{x}(t) - A_0(t)\,x(t) = f(t)$$

which corresponds to the operator $L := \partial - A_0$.

```
OL := Prod[Diff] - Prod[A0]
```

Our goal is to construct a solution operator $H := H_1 \cdot \int \cdot H_2$ with the unknowns H_1 and H_2 such that $L \cdot H = 1$.

```
OH := Prod[H1, Int, H2]
```

We apply the reduction system to $L \cdot H - 1$ and extract its left coefficients.

```
ApplyRules[Prod[OL, OH] - Prod[], RedSys]

-Prod[] + Prod[mul[H1, H2]] + Prod[Diff[H1], Int, H2] - Prod[mul[A0, H1], Int, H2]
```

```
% // Coefficients // TableForm

Prod[]            -1 + mul[H1, H2]
Prod[Int, H2]     Diff[H1] - mul[A0, H1]
```

Then by coefficient comparison we obtain the conditions

$$H_1 H_2 = 1 \qquad \text{and} \qquad \partial H_1 - A_0 H_1 = 0.$$

A solution for this system is obtained by choosing an invertible Φ s.t. $\partial\Phi - A_0\Phi = 0$. Taking $H_1 = \Phi$ and $H_2 = \Phi^{-1}$ we obtain

$$H = \Phi \cdot \int \cdot \Phi^{-1}.$$

```
Diff[Φ] := mul[A₀, Φ]
OH = (OH /. {H₁ → Φ, H₂ → inv[Φ]});
```

We can also check the operator $H = \Phi \cdot \int \cdot \Phi^{-1}$ satisfies the identity $L \cdot H = 1$.

```
ApplyRules[Prod[OL, OH] - Prod[], RedSys]
```

0

The equation $L\,x = f$ from example above is equivalent to the equation $(H \cdot L)\,x = H\,f$. Let H and Φ be defined as above. Applying the reduction system, we can easily find $1 - \Phi\,E\,\Phi^{-1} \cdot E$ as the irreducible form of $H \cdot L$.

```
ApplyRules[Prod[OH, OL], RedSys]
```

```
Prod[] - Prod[mul[Φ, Eval[inv[Φ]]], Eval]
```

Let us define $P := \Phi\,E\,\Phi^{-1} \cdot E$ where P is a projector.

```
OP := Prod[mul[Φ, Eval[inv[Φ]]], Eval]
ApplyRules[Prod[OP, OP] - OP, RedSys]
```

0

Considering the operator P allows us to write $(H \cdot L)\,x = H\,f$ as the following recurrence equation.

$$x = P\,x + H\,f$$

Ex.9) Differential time delay systems (method of steps)

Consider the differential time delay system

$$\dot{x}(t) - A_0(t)\,x(t) + A_1(t)\,x(t-h) = f(t),$$

which corresponds to the operator $R := L + S$
where the operator L is defined like above and $S := -A_1 \cdot \delta$.

```
OS := -Prod[A₁, δ]
OR := OL + OS
```

Our goal is to reproduce “method of steps” for general solution of this system. We apply the reduction system to the operator $H \cdot R$.

```
ApplyRules[Prod[OH, OR], RedSys]
```

```
Prod[] - Prod[mul[Φ, Eval[inv[Φ]]], Eval] -
 Prod[Φ, S[1, h], Int, mul[inv[S[1, -h][Φ]], S[1, -h][A₁]]] +
 Prod[Φ, Eval, S[1, h], Int, mul[inv[S[1, -h][Φ]], S[1, -h][A₁]]]
```

The identity $H \cdot R = H \cdot L + H \cdot S = 1 - P + H \cdot S = 1 - (P - H \cdot S)$ allows us to define an operator G as follows.

$$G := P - H \cdot S$$

```
OG := OP - Prod[OH, OS]
```

We can also check for the operator G the identity $H \cdot R = 1 - G$ holds.

```
ApplyRules[Prod[OH, OR] - (Prod[] - OG), RedSys]
```

0

Considering the operator G allows us to rewrite $(H\cdot R)\,x = H f$ as the following recurrence equation.

$$x = G x + H f$$

Ex.10) Variation of constants with rectangular coefficients

Consider the inhomogeneous differential system

$$A_1(t)\,\dot{x}(t) - A_0(t)\,x(t) = f(t),$$

which corresponds to the operator $L := A_1 \cdot \partial - A_0$ with rectangular coefficients A_0 and A_1.

```
OL := Prod[A₁, Diff] - Prod[A₀]
```

Our goal is to construct a solution operator $H = H_1 \cdot \int \cdot H_2$ with unknowns H_1 and H_2 such that the identity $L\cdot H = 1$ holds.

```
OH := Prod[H₁, Int, H₂]
```

We apply the reduction system to $L\cdot H$ and extract its left coefficients.

```
ApplyRules[Prod[OL, OH], RedSys]
```

```
Prod[mul[A₁, H₁, H₂]] - Prod[mul[A₀, H₁], Int, H₂] + Prod[mul[A₁, Diff[H₁]], Int, H₂]
```

```
% // Coefficients // TableForm
```

```
Prod[]           mul[A₁, H₁, H₂]
Prod[Int, H₂]    -mul[A₀, H₁] + mul[A₁, Diff[H₁]]
```

Comparing coefficients in $L\cdot H = 1$ yields

$$A_1\,H_1\,H_2 = 1 \quad \text{and} \quad A_1\,\partial H_1 - A_0\,H_1 = 0.$$

A solution is obtained by choosing Θ and $\tilde{\Theta}$ s.t. $A_1\,\Theta\,\tilde{\Theta} = 1$ and $A_1\,\partial\,\Theta - A_0\,\Theta = 0$. Taking $H_1 = \Theta$ and $H_2 = \tilde{\Theta}$ we obtain

$$H = \Theta \cdot \int \cdot \tilde{\Theta}.$$

```
mul[a___, A₁, Θ, ΘΘ, b___] := mul[a, b]
mul[a___, A₁, Diff[Θ], b___] := mul[a, A₀, Θ, b]
OH = (OH /. {H₁ → Θ, H₂ → ΘΘ});
```

We can easily check the operator $H = \Theta \cdot \int \cdot \tilde{\Theta}$ satisfies the identity $L\cdot H = 1$.

```
ApplyRules[Prod[OL, OH] - Prod[], RedSys]
```

```
0
```